

Abnormal Heartbeat Detector
Proyek Mikrokontroller 8051



Universitas Indonesia

Kelompok : Blue Pill

Ramadhan Kalih Sewu (1806148826)

Qisas Tazkia Hasanudin (1806200210)

Bryan Oliver (1806200305)

Farid Muhammad Rahman (1806148662)

FAKULTAS TEKNIK
DEPARTEMEN TEKNIK ELEKTRO

DEPOK 14 May 2020

I. Deskripsi sistem

Pemilihan Masalah :


Berdasarkan topik COVID-19, yang dapat berupa pencegahan, penanganan, dan pemantauan, kelompok kami mengambil satu topik yang berupa pemantauan. Hal ini didasari oleh penemuan berita yang dilansir oleh Suara.com pada 9 April 2020. 'Infeksi virus dapat mengganggu aliran darah ke jantung, menyebabkan detak jantung tidak teratur dan gagal jantung.' Dikutip dari Risna Halidi | Dini Afrianti Efendi <https://www.suara.com/health/2020/04/09/120000/studi-ungkap-1-dari-5-pasien-covid-19-alami-kerusakan-jantung>.

Gambar I: Pemicu Masalah

Studi Ungkap 1 dari 5 Pasien Covid-19 Alami Kerusakan Jantung

Risna Halidi | Dini Afrianti Efendi

Kamis, 09 April 2020 | 12:00 WIB



Ilustrasi jantung manusia (Shutterstock).

Peneliti berasumsi virus menyerang langsung sel jantung atau anggota tubuh lain yang pada akhirnya malah memiliki efek merusak jantung.

Tujuan Program :

Program ini dirancang untuk mendeteksi detak jantung dari sebuah sinyal square wave. Program akan memberi tahu kepada pengguna mengenai kondisi detak jantung berdasarkan umurnya. Tidak berhenti disitu, alat ini memiliki counter untuk menghitung berapa banyak penderita yang memiliki penyakit jantung. Sehingga dapat digunakan untuk statistik para tenaga medis dalam bekerja dilapangan dalam mengecek kondisi Jantung penderita Covid.

Sirkuit Embedded :

Rangkaian dibuat dengan prinsip simplicity, ease of use, dan interactive. Kami menggunakan seluruh port yang tersedia pada 8051 dengan upaya memanfaatkan seluruh Output yang ada sehingga data yang ditampilkan lengkap.

Input :

- Keypad
- Sensor (Square Wave Signal)

Output :

- LCD 16x2
- 2 BCD Seven Segment
- Serial Communication
- Buzzer
- Indikator LED

Seiring dengan jalannya rangkaian, akan dipandu oleh LCD 16x2 serta Serial Communication, pengguna tidak perlu khawatir dengan cara menggunakannya, karena setiap langkah telah dijelaskan secara perlahan.


Cara Penggunaan Program :

Saat device baru dinyalakan, LCD dan serial monitor akan menampilkan boot-up text yang menunjukkan nama kelompok dan nama proyek. Lalu, program akan meminta user untuk memasukkan usianya melalui keypad, dengan usia minimal 00 dan maksimal 99. Setelah itu, program akan meminta user untuk meletakkan jarinya pada sensor, sambil menunggu sampai sensor mendeteksi perubahan cahaya, yang berasal dari tekanan darah pada jari user. Jika terdeteksi, device akan menghitung berapa banyak detak jantung yang terdeteksi oleh sensor selama 5 detik. Hasil ini akan dikalkulasi, sehingga didapatkan kecepatan detak jantung dalam satuan BPM.

Cara Kerja Program :

Untuk menentukan kualitas detak jantung user, telah diberikan parameter setiap umur yang memungkinkan pada Program yang di tulis di ROM. Program akan mengecek apabila Heartbeat diluar Lower Bound dan Upper Bound. Apabila kondisi tersebut terjadi maka Tampilkan BAD. Program mengambil angka Ideal, lalu mengecek -10 (Sebagai Lower Bound) dan +10 (Sebagai Upper Bound). Jika Heartbeat ada dalam range ideal, maka tampilkan GOOD. Kondisi lainnya tampilkan FAIR.

Gambar II : Rangkaian Sistem Embedded dalam Proteus



Gambar III : Data Referensi Heart Rate

| Age | Heart Rate | | |
|---------------|-------------|-------|-------------|
| | Lower Bound | Ideal | Upper Bound |
| < 1 Years | 80 | 120 | 160 |
| 1 - 2 Years | 80 | 105 | 130 |
| 3 - 4 Years | 80 | 100 | 120 |
| 5 - 6 Years | 75 | 95 | 115 |
| 7- 9 Years | 70 | 85 | 100 |
| 10 - 17 Years | 60 | 80 | 100 |
| > 18 Years | 40 | 60 | 80 |

Addressing Modes :

a) Register Addressing

```
MOV    A, R6           ;MSB Saved in R6
```

b) Indexed Addressing

```
MOVC    A, @A+DPTR
```


c) Immediate Addressing

```
MOV     B,    #LOW(-8H)    ;Delay for -8H
```


d) Register Indirect

```
MOV A, @R0
```

II. Flowchart sistem



Output Process



III. Source code

Repository : https://github.com/ramdanks/Abnormal_Heartbeat

```
1  ; ===== ( ABOUT PROGRAM ) ===== ;
2  ;About      : Abnormal Heartbeat Detector
3  ;Author     : Kelompok. Blue Pill
4  ;Anggota    : - Bryan Oliver
5  ;           : - M. Farid Rahman
6  ;           : - Ramadhan Kalih Sewu
7  ;           : - Qisas Tazkia Hasanuddin
8  ;Matkul     : Sistem Berbasis Komputer
9  ;Hal        : Proyek Mikrokontroler 8051
10
11 ; ===== ( DOCUMENTATION ) ===== ;
12 ;Repository  : https://github.com/ramdanks/Abnormal_Heartbeat
13 ;License     : GNU GPL-3.0
14
15 ; ===== ( SIMULATION ) ===== ;
16 ;Tools       : Proteus 8 Professional
17 ;XTAL Freq.  : 12 MHz
18 ;MP Freq.    : 12 MHz
19 ;Baud Rate SRL : 31250 BPS
20 ;Com. Interface : Serial, LCD 16x2
21
22 KEYPAD      EQU    P3
23 KP_A        EQU    P3.0
24 KP_B        EQU    P3.2
25 KP_C        EQU    P3.3
26 KP_D        EQU    P3.4
27 KP_1        EQU    P3.5
28 KP_2        EQU    P3.6
29 KP_3        EQU    P3.7
30 KP_ST       EQU    0H
31
32 BAUD_RATE    EQU    LOW(-1)      ;31250 Baud Rate
33
34 SEVSEG       EQU    P1
35 S_LSB        EQU    07H
36 S_MSB        EQU    06H
37
38 HRT_SNS      EQU    P2.0
39
40 BUZZ         EQU    P2.1
41 LEDR         EQU    P2.2
42 LEDY         EQU    P2.3
43 LEDG         EQU    P2.4
44
45 RS           EQU    P2.5
46 RW           EQU    P2.6
47 ENB          EQU    P2.7
48
49 RESULTBPM_ASC EQU    30H
50 RESULTBPM     EQU    55H
51 INPUTAGE      EQU    60H
52
53 LCD16         EQU    P0
54 LCD_ACT       EQU    038H
55 LCD_ON        EQU    00FH
56 LCD_CLR       EQU    001H
57 LCD_LN1       EQU    080H
58 LCD_LN2       EQU    0C0H
59 LCD_LN        EQU    000H
60
61
```



```

--
61 ;=====! (DATA) !=====;
62         ORG         300H
63
64 NUM:          DB      31H,32H,33H,34H,35H,36H,37H,38H,39H,00H,30H,00H
65
66 STR_KLM:      DB      'BLUE PILL', 0
67 STR_TTL1:     DB      'ABN. HEARTBEAT', 0
68 STR_TTL2:     DB      'DETECTOR', 0
69 STR_INP:      DB      'Please ENTER', 0
70 STR_AGE:      DB      'Your AGE: ', 0
71 STR_DTT1:     DB      'FINGER SENSOR', 0
72 STR_DTT2:     DB      'WAITING...', 0
73 STR_LST1:     DB      'Keep FINGER', 0
74 STR_LST2:     DB      'On The SENSOR', 0
75 STR_HRT:      DB      'HB: ', 0
76 STR_BPM:      DB      ' (BPM)', 0
77 STR_SUM:      DB      'Result: ', 0
78 STR_GOOD:     DB      'GOOD', 0
79 STR_FAIR:     DB      'FAIR', 0
80 STR_BAD:      DB      'BAD', 0
81 STR_NL:       DB      '\r\n', 0
82
83 STR_CNS_BRD:   DB      '=====', 0
84 STR_CNS_OK:    DB      'Press (*) [Ok] to Continue...', 0
85 STR_CNS_RST:   DB      'Press (#) [Reset] to Clear Input...', 0
86 STR_CNS_CNF:   DB      'Age Confirm, Put your Finger on the Sensor!', 0
87 STR_CNS_LST:   DB      'Keep Finger on the Sensor...', 0
88
89 ;HeartRate          LowB   Ideal   UppB
90 YEAR_RATE:
91 YEAR0:            DB      80,   120,  160
92 YEAR1:            DB      80,   105,  130
93 YEAR2:            DB      80,   105,  130
94 YEAR3:            DB      80,   100,  120
95 YEAR4:            DB      80,   100,  120
96 YEAR5:            DB      75,   95,   115
97 YEAR6:            DB      75,   95,   115
98 YEAR7:            DB      70,   85,   100
99 YEAR8:            DB      70,   85,   100
100 YEAR9:           DB      70,   85,   100
101 YEAR10:          DB      60,   80,   100
102 YEAR11:          DB      60,   80,   100
103 YEAR12:          DB      60,   80,   100
104 YEAR13:          DB      60,   80,   100
105 YEAR14:          DB      60,   80,   100
106 YEAR15:          DB      60,   80,   100
107 YEAR16:          DB      60,   80,   100
108 YEAR17:          DB      60,   80,   100
109 YEAR18:          DB      40,   60,   80

```

```

~~~
104 ;=====! (MACROS) !=====;
105 LCD_CLEAR MACRO
106     MOV     A, #LCD_CLR
107     ACALL   LCD_CMD
108 ENDM
109
110 LCD_NL MACRO
111     MOV     A, #LCD_LN2
112     ACALL   LCD_CMD
113 ENDM
114
115 LCD_PRINT MACRO STR_PARAM
116     MOV     DPTR, #STR_PARAM
117     ACALL   LCD_WRITE
118 ENDM
119
120 LCD_PRINT_RAM MACRO STR_PARAM
121     MOV     R0, #STR_PARAM
122     ACALL   LCD_WRITERAM
123 ENDM
124
125 LCD_PRINT2 MACRO STR_P1, STR_P2
126     MOV     DPTR, #STR_P1
127     ACALL   LCD_WRITE
128     LCD_NL
129     MOV     DPTR, #STR_P2
130     ACALL   LCD_WRITE
131 ENDM
132
133 SERIAL_NL MACRO
134     MOV     DPTR, #STR_NL
135     ACALL   SSTR
136 ENDM
137
138 SERIAL_PRINT MACRO STR_PARAM
139     MOV     DPTR, #STR_PARAM
140     ACALL   SSTR
141 ENDM

143 SERIAL_PRINT_RAM MACRO MEM_PARAM
144     MOV     A, MEM_PARAM
145     ACALL   WRITE_SERIAL_RAM
146 ENDM
147
148 SERIAL_PRINT_NL MACRO STR_PARAM
149     SERIAL_PRINT STR_PARAM
150     SERIAL_NL
151 ENDM

```

```

---
153 ;=====! (SUBROUTINE) !=====;
154     ORG     500H
155
156 SHOW7:        ;Show 2 Digit Decimal in 7 Segment
157     MOV     A, R6            ;MSB Saved in R6
158     MOV     B, #4            ;Shift Left 4 Times
159     SL:      RL      A        ;4 Bit for MSB, 4 Bit for LSB
160             DJNZ     B, SL
161     ADD     A, R7            ;LSB added in last 4 bit
162     MOV     SEVSEG, A        ;Move 8 Bit Data (4 Bit MSB, 4 Bit LSB) to Port Seven Segment
163     RET
164
165 SSTR:
166     CLR     A
167     MOVC    A, @A+DPTR
168     JZ      S_RET
169     ACALL   WRITE_SERIAL_RAM
170     INC     DPTR
171     SJMP    SSTR
172     S_RET:  RET
173
174 WRITE_SERIAL_RAM:
175     MOV     SCON, #50H
176     MOV     TMOD, #20H
177     MOV     TH1, #BAUD_RATE
178     SEND:   SETB    TR1
179             MOV     SBUF, A
180     WAIT:   JNB     TI, WAIT
181             CLR     TR1
182             CLR     TI
183             MOV     TL1, #0
184             MOV     TH1, #0
185             RET
186
187 DELAY:
188     MOV     TMOD, #01H        ;Timer 0 16 Bit
189     MOV     TH0, B            ;Parameter B is used to give Timer0 Initiation Value
190     SETB    TR0              ;Start Timer 0
191     HERE:   JNB     TF0, HERE  ;Wait until Overflow
192             CLR     TR0        ;Stop Timer
193             CLR     TF0        ;Clear Overflow
194             RET
195
196 FIX_DELAY:
197     MOV     A, #10H           ;Nested Loop Delay 10 Times!
198     D_AGAIN:MOV B, #LOW(-0FFH) ;Set Timer0 Initiation Value
199             ACALL   DELAY      ;Delay
200             DJNZ     A, D_AGAIN ;Delay Again for 10 Times
201     RET
202

```

```

203 ;=====! (LCD SUBROUTINE) !=====;
204
205 LCD_INIT:      ;Do the necessary thing to make LCD working
206               MOV     A,      #LCD_ACT      ;Activate LCD with 2 Lines 5x7 Matrix
207               ACALL    LCD_CMD
208               MOV     A,      #LCD_ON       ;Turn on LCD with Cursor
209               ACALL    LCD_CMD
210               MOV     A,      #LCD_LN1      ;Force cursor to 1st Line
211               ACALL    LCD_CMD
212               RET
213
214 LCD_WRITE:      ;Write String to LCD (from Cursor Location)
215               CLR     A                  ;Clear A
216               MOVC    A,      @A+DPTR      ;Load next Char
217               JZ      LCD_RET            ;if Content of A is NULL, then we're done printing
218               ACALL    LCD_TXT            ;Show to LCD using write command
219               INC     DPTR                ;Proceed to next Char
220               SJMP    LCD_WRITE           ;Loop again until Reaches NULL Terminating Char
221               LCD_RET: RET
222
223 LCD_WRITERAM:   ;Write Text to LCD from Text Located in RAM
224               ;The reason our project use this, is because we cannot use DPTR for accessing RAM
225               ;So we create alternative by using R0 as a parameter.
226               MOV     A, @R0              ;R0 is parameter for memory location of string (Like DPTR)
227               JZ      LCD_RET            ;if Content of A is NULL, then we're done printing
228               ACALL    LCD_TXT            ;Show to LCD using write command
229               INC     R0                  ;Proceed to next Char
230               SJMP    LCD_WRITERAM        ;Loop again until Reaches NULL Terminating Char
231
232 LCD_CMD:        ;Give Command to LCD
233               MOV     LCD16, A            ;A is parameter for LCD Command
234               CLR     RS                  ;RS set to 0
235               CLR     RW                  ;RW set to 0
236               SETB    ENB                 ;E set to 1 (for enabling command)
237               CLR     ENB                 ;Don't forget to clear E (command only trigger once)
238               MOV     B,      #LOW(-8H)   ;Delay for -8H
239               ACALL    DELAY
240               RET
241
242 LCD_TXT:        ;Sending Text to LCD
243               MOV     LCD16, A            ;A is parameter for char in ASCII (to write in LCD)
244               SETB    RS                  ;RS set to 1
245               CLR     RW                  ;RW set to 0
246               SETB    ENB                 ;E set to 1 (for enabling write command)
247               CLR     ENB                 ;Don't forget to clear E (write command only trigger once)
248               MOV     B,      #LOW(-8H)   ;Delay for -8H
249               ACALL    DELAY
250               RET

```

```

252 ;=====! (KEYPAD SUBROUTINE) !=====;
253
254 KP_READ:
255         SETB     KEYPAD
256         MOV      DPTR,#NUM
257
258         MOV KEYPAD, #11111110B
259 ONE:    MOV KP_ST,#00H
260         JNB KP_1, LCD_WRITE_KP
261 TWO:    INC KP_ST
262         JNB KP_2, LCD_WRITE_KP
263 THR:    INC KP_ST
264         JNB KP_3, LCD_WRITE_KP
265
266         MOV KEYPAD,#11111011B
267 FOUR:   INC KP_ST
268         JNB KP_1, LCD_WRITE_KP
269 FIVE:   INC KP_ST
270         JNB KP_2, LCD_WRITE_KP
271 SIX:    INC KP_ST
272         JNB KP_3, LCD_WRITE_KP
273
274         MOV KEYPAD,#11110111B
275 SVN:    INC KP_ST
276         JNB KP_1, LCD_WRITE_KP
277 EGT:    INC KP_ST
278         JNB KP_2, LCD_WRITE_KP
279 NINE:   INC KP_ST
280         JNB KP_3, LCD_WRITE_KP
281
282         MOV KEYPAD,#11101111B
283 STAR:   INC KP_ST
284 ZERO:   INC KP_ST
285         JNB KP_2, LCD_WRITE_KP
286 HASH:   INC KP_ST
287
288 NONE:   SJMP KP_READ
289
290 LCD_WRITE_KP:
291         MOV      A, KP_ST
292         MOVC     A,@A+DPTR
293         ACALL    WRITE_SERIAL_RAM
294         ACALL    LCD_TXT
295
296 KP_WAIT:
297         JNB      KP_1, KP_WAIT
298         JNB      KP_2, KP_WAIT
299         JNB      KP_3, KP_WAIT
300         RET
301

```

```

302 ;=====! (STATE ROUTINE) !=====;
303
304 HELLO:
305         LCD_CLEAR
306         SERIAL_PRINT_NL STR_KLM
307         LCD_PRINT      STR_KLM
308         ACALL          FIX_DELAY
309         LCD_CLEAR
310         SERIAL_PRINT_NL STR_TTL1
311         SERIAL_PRINT_NL STR_TTL2
312         SERIAL_NL
313         LCD_PRINT2      STR_TTL1, STR_TTL2
314         ACALL  FIX_DELAY
315         RET
316
317 INPUT:
318         SERIAL_PRINT_NL STR_CNS_BRD
319         SERIAL_PRINT  STR_AGE
320         LCD_CLEAR
321         LCD_PRINT2      STR_INP, STR_AGE
322         ACALL  KP_READ
323         SUBB   A, #30
324         MOV    B, #10
325         MUL    AB
326         MOV    B, A
327         ACALL  KP_READ
328         SUBB   A, #30
329         ADD    A, B
330         MOV    INPUTAGE, A
331         SERIAL_NL
332         SERIAL_PRINT_NL STR_CNS_OK
333         SERIAL_PRINT_NL STR_CNS_RST
334
335         MOV    KEYPAD, #11101111B
336         WAIT_KEY:
337         JNB    KP_3, INPUT          ;if Reset Button is Pressed ask Input Again!
338         JB     KP_1, WAIT_KEY       ;Wait Until OK Button is Pressed
339         RET
340
341 KEEPMMSG:
342         SERIAL_PRINT_NL STR_CNS_LST
343         LCD_CLEAR
344         LCD_PRINT2      STR_LST1, STR_LST2
345         RET

```

```

345
346 LISTEN:      ;Process Heartbeat Sensor
347             LCD_CLEAR
348             LCD_PRINT2      STR_DTT1, STR_DTT2
349             SERIAL_PRINT_NL STR_CNS_CNF
350             MOV     R3, #0      ;R3 is used for Counting Time
351             LOWPULSE:  JNB     TF0, LOW_DETECT      ;Low Signal Detection
352                     CLR     TF0
353                     INC     R3
354                     CJNE    R3, #152, LOW_DETECT    ;Continue Detection if < 10 SEC
355                     SJMP    DONE_10SEC
356             LOW_DETECT: JNB     HRT_SNS, LOWPULSE    ;Wait until Low Pulse
357             HIGHPULSE: JNB     TF0, HIGH_DETECT     ;High Signal Detection
358                     CLR     TF0
359                     INC     R3
360                     CJNE    R3, #152, HIGH_DETECT   ;Continue Detection if < 10 SEC
361                     SJMP    DONE_10SEC
362             HIGH_DETECT: JB      HRT_SNS, HIGHPULSE   ;Wait until High Pulse
363             CJNE    A, #0, NOWCOUNTING              ;First Heartbeat Pulse
364             ACALL   KEEPMSG                          ;Show MSG to LCD
365             MOV     A, #0      ;A is used for counting BPM
366             MOV     TH0, #0     ;Set Timer Initiation from 0
367             MOV     TL0, #0
368             SETB    TR0          ;Turn on Timer on first Pulse
369             NOWCOUNTING: INC    A      ;Increment Beat every Wave
370                     SJMP    LOWPULSE    ;Detect Again!
371             DONE_10SEC:  CLR     TR0     ;Stop Timer
372                     MOV     B, #6      ;Multiply By 6
373                     MUL     AB         ;Because we Only Scan 10 SEC
374                     MOV     RESULTBPM, A ;Save BPM
375             RET
376
377 PROCESS_RESULT:
378             SERIAL_PRINT      STR_SUM
379             LCD_PRINT         STR_SUM
380             MOV     A, INPUTAGE
381             CJNE    A, #18, CHECKAGE
382             SJMP    ABOVE18Y
383             CHECKAGE: JC      PROC
384             ABOVE18Y: MOV     A, #18
385             PROC:    MOV     B, #3
386                     MUL     AB
387                     MOV     B, A
388                     MOV     DPTR, #YEAR_RATE
389
390             LOWERBOUND_CHECK:
391             MOVC     A, @A+DPTR
392             SUBB     A, RESULTBPM
393             JNC      TRIGGERBAD
394
395             UPPERBOUND_CHECK:
396             MOV     A, B
397             INC     A
398             INC     A
399             MOVC     A, @A+DPTR
400             SUBB     A, RESULTBPM
401             JC      TRIGGERBAD
402

```

```

403 CHECK_GOOD:
404 INC B
405
406 LOWERBOUND_GOOD:
407 MOV A, B
408 MOVC A, @A+DPTR
409 SUBB A, #10
410 SUBB A, RESULTBPM
411 JNC TRIGGERFAIR
412
413 UPPERBOUND_GOOD:
414 MOV A, B
415 MOVC A, @A+DPTR
416 ADD A, #10
417 SUBB A, RESULTBPM
418 JC TRIGGERFAIR
419
420 TRIGGERGOOD: ACALL GOOD
421 RET
422 TRIGGERFAIR: ACALL FAIR
423 RET
424 TRIGGERBAD: ACALL BAD
425 RET
426
427 CONVERT: ;Convert BPM from 1 Block RAM to 3 ASCII char
428 MOV A, RESULTBPM
429 MOV B, #10
430 DIV AB
431 MOV 32H, B
432 MOV B, #10
433 DIV AB
434 MOV 31H, B
435 MOV 30H, A
436 ;Change Based Number to ASCII
437 ORL 30H, #30H
438 ORL 31H, #30H
439 ORL 32H, #30H
440 SERIAL_PRINT STR_HRT
441 SERIAL_PRINT_RAM 30H
442 SERIAL_PRINT_RAM 31H
443 SERIAL_PRINT_RAM 32H
444 SERIAL_PRINT_NL STR_BPM
445 RET
446
447 SHOWBPM: ;Show Result HeartRate (BPM), to LCD
448 LCD_CLEAR
449 LCD_PRINT STR_HRT
450 LCD_PRINT_RAM RESULTBPM_ASC
451 LCD_PRINT STR_BPM
452 LCD_NL
453 RET

```



```

455 INIT_PORT:      ;Initialize Port when First Bootup
456 CLR    BUZZ      ;Turn off Buzzer
457 MOV    S_MSB, #0  ;Clear MSB Counter
458 MOV    S_LSB, #0  ;Clear LSB Counter
459 ACALL  SHOW7      ;Show Counter
460 CLR    HRT_SNS    ;Clear Port Sensor
461 SETB   KEYPAD
462 RET
463
464 GOOD:           ;Heartrate is Ideal
465 CLR     LEDG      ;Turn on Green LED
466 LCD_PRINT    STR_GOOD
467 SERIAL_PRINT_NL STR_GOOD
468 RET
469
470 FAIR:           ;Heartrate is Safe
471 CLR     LEDY      ;Turn on Yellow LED
472 LCD_PRINT    STR_FAIR
473 SERIAL_PRINT_NL STR_FAIR
474 RET
475
476 BAD:            ;Heartrate is Outside Safe Range
477 CLR     LEDR      ;Turn on Red LED
478 SETB   BUZZ
479 LCD_PRINT    STR_BAD
480 SERIAL_PRINT_NL STR_BAD
481 CJNE    R7, #9, INC_LSB      ;Increment Counter and Show Decimal Number
482 CJNE    R6, #9, INC_MSB
483 SJMP    B_RET
484 INC_MSB:      INC     R6
485               MOV     R7, #0
486               SJMP    B_RET
487 INC_LSB:      INC     R7
488 B_RET:       ACALL  SHOW7  ;Show Counter in 7 Segment Form
489             RET
490
491 WAIT_RESET:
492 MOV     KEYPAD, #11101111B
493 JB      KP_3, WAIT_RESET    ;Loop until Reset Button is Pressed!
494 CLR     BUZZ
495 SETB    LEDR
496 SETB    LEDG
497 SETB    LEDY
498 SERIAL_NL
499 RET
500

```

```
501 ;=====! (MAIN) !=====;
502         ORG     00H
503
504 ;Inisiasi
505 ACALL    INIT_PORT
506 ACALL    LCD_INIT
507 ACALL    HELLO
508
509 MAIN:    ACALL    INPUT
510          ACALL    LISTEN
511          ACALL    CONVERT
512          ACALL    SHOWBPM
513          ACALL    PROCESS_RESULT
514          ACALL    WAIT_RESET
515          SJMP     MAIN
516
517 HALT:    SJMP     HALT
```

IV. Link video presentasi

Presentasi + Penjelasan : <https://youtu.be/0wvor58aK6s>

Simulasi : <https://youtu.be/rC1mJqsGn1Y>

V. Referensi

Pemicu Masalah :

[https://www.suara.com/health/2020/04/09/120000/studi-ungkap-1-dari-5-pasien-covid-19-
alami-kerusakan-jantung](https://www.suara.com/health/2020/04/09/120000/studi-ungkap-1-dari-5-pasien-covid-19-alami-kerusakan-jantung)

Heart Rate Berdasarkan Umur :

<https://www.medicalnewstoday.com/articles/235710#normal-resting-heart-rate>

https://www.freeprintablemedicalforms.com/preview/Resting_Heart_Rate_Chart

LCD 16x2 Pinout Diagram and Commands :

<https://www.electronicsforu.com/resources/learn-electronics/16x2-lcd-pinout-diagram>

Program Repository :

https://github.com/ramdanks/Abnormal_Heartbeat